
AMath 483/583 — Lecture 22 — May 16, 2011

Today:
• Gauss-Seidel and SOR iterative methods
• Totalview debugging

Next:
• GPUs, Python

Read: Class notes and references

Jacobi iteration

(Ui−1 − 2Ui + Ui+1) = −∆x2f(xi)

Solve for Ui:

Ui =
1
2

(
Ui−1 + Ui+1 + ∆x2f(xi)

)
.

Note: With no heat source, f(x) = 0,
the temperature at each point is average of neighbors.

Suppose U [k] is a approximation to solution. Set

U
[k+1]
i =

1
2

(
U

[k]
i−1 + U

[k]
i+1 + ∆x2f(xi)

)
for i = 1, 2, . . . , n.

Repeat for k = 0, 1, 2, . . . until convergence.

Can be shown to converge (eventually... very slow!)

Jacobi iteration in Fortran

uold = u ! starting values before updating

do iter=1,maxiter

dumax = 0.d0

do i=1,n
u(i) = 0.5d0*(uold(i-1) + uold(i+1) + dx**2*f(i))
dumax = max(dumax, abs(u(i)-uold(i)))
enddo

! check for convergence:
if (dumax .lt. tol) exit

uold = u ! for next iteration
enddo

Note: we must use old value at i− 1 for Jacobi.

Otherwise we get the Gauss-Seidel method.

Gauss-Seidel iteration in Fortran

do iter=1,maxiter

dumax = 0.d0

do i=1,n
uold = u(i)
u(i) = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))
dumax = max(dumax, abs(u(i)-uold))
enddo

! check for convergence:
if (dumax .lt. tol) exit

enddo

Note: Now u(i) depends on value of u(i-1) that has already
been updated for previous i.

Good news: This converges about twice as fast as Jacobi!

But... loop carried dependence! Cannot parallelize so easily.

Red-black ordering

We are free to write equations of linear system in any order...
reordering rows of coefficient matrix, right hand side.

Can also number unknowns of linear system in any order...
reordering elements of solution vector.

Red-black ordering: Iterate through points with odd index first
(i = 1, 3, 5, . . .) and then even index points (i = 2, 4, 6, . . .).

Then all black points can be updated in any order,
all red points can then be updated in any order.

Same asymptotic convergence rate as natural ordering.

Red-Black Gauss-Seidel

do iter=1,maxiter
dumax = 0.d0

! UPDATE ODD INDEX POINTS:
!$omp parallel do reduction(max : dumax) &
!$omp private(uold)
do i=1,n,2

uold = u(i)
u(i) = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))
dumax = max(dumax, abs(u(i)-uold))
enddo

! UPDATE EVEN INDEX POINTS:
!$omp parallel do reduction(max : dumax) &
!$omp private(uold)
do i=2,n,2

uold = u(i)
u(i) = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))
dumax = max(dumax, abs(u(i)-uold))
enddo

! check for convergence:
if (dumax .lt. tol) exit
enddo

Gauss-Seidel method in 2D

If ∆x = ∆y = h:

1
h2

(Ui−1,j + Ui+1,j + Ui,j−1 + Ui,j+1 − 4Ui,j) = −f(xi, yj).

Solve for Ui,j and iterate:

u
[k+1]
i,j =

1
4

(u[k+1]
i−1,j + u

[k]
i+1,j + u

[k+1]
i,j−1 + u

[k]
i,j+1 − h2fi,j)

Again no need for matrix A.

Note: Above indices for old and new values assumes we iterate
in the natural row-wise order.

Gauss-Seidel in 2D

Updating point 7 for example (u32):

Depends on new values at points 6 and 3, old values at points
7 and 10.

U
[k+1]
32 =

1
4

(U [k+1]
22 + U

[k]
42 + U

[k+1]
21 + U

[k]
41 + h2f32)

Red-black ordering in 2D

Again all black points can be updated in any order:
New value depends only on red neighbors.

Then all red points can be updated in any order:
New value depends only on black neighbors.

SOR method

Gauss-Seidel move solution in right direction but not far enough
in general.

Iterates “relax” towards solution.

Successive Over-Relaxation (SOR):

Compute Gauss-Seidel approximation and then go further:

UGS
i =

1
2

(U [k+1]
i−1 + U

[k]
i+1 + ∆x2f(xi))

U
[k+1]
i = U

[k]
i + ω(UGS

i − U
[k]
i )

where 1 < ω < 2.

Optimal omega (For this problem): ω = 2− 2π∆x.

Convergence rates

Red-Black SOR in 1D

do iter=1,maxiter
dumax = 0.d0

! UPDATE ODD INDEX POINTS:
!$omp parallel do reduction(max : dumax) &
!$omp private(uold, ugs)
do i=1,n,2

uold = u(i)
ugs = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))
u(i) = uold + omega*(ugs-uold)
dumax = max(dumax, abs(u(i)-uold))
enddo

! UPDATE EVEN INDEX POINTS:
!$omp parallel do reduction(max : dumax) &
!$omp private(uold, ugs)
do i=2,n,2

uold = u(i)
ugs = 0.5d0*(u(i-1) + u(i+1) + dx**2*f(i))
u(i) = uold + omega*(ugs-uold)
dumax = max(dumax, abs(u(i)-uold))
enddo

! check for convergence...

Note that uold, ugs must be private!

Totalview debugger

http://www.roguewave.com/products/totalview-family

Student versions available — see email to class.

